TVM Functions in EXCEL

Order of Variables = (Rate, Nper, Pmt, Pv, Fv,Type, Guess)

Future Value = FV(Rate,Nper,Pmt,PV,Type)
Present Value = PV(rate,nper,pmt,fv,type)
No. of Periods = NPER(rate, pmt, pv, fv, type)
Payment = PMT(rate,nper,pv,fv,type)
Rate = RATE(nper,pmt,pv,fv,type,guess)

Rate is the interest rate per period. For example, if you obtain an automobile loan at a 10 percent annual interest rate and make monthly payments, your interest rate per month is 10%/12, or 0.83%. You would enter 10%/12, or 0.83%, or 0.0083, into the formula as the rate.

Nper is the total number of payment periods in an annuity. For example, if you get a four-year car loan and make monthly payments, your loan has 4*12 (or 48) periods. You would enter 48 into the formula for nper.

Pmt is the payment made each period and cannot change over the life of the annuity. Typically, pmt includes principal and interest but no other fees or taxes. For example, the monthly payments on a $10,000, four-year car loan at 12 percent are $263.33. You would enter -263.33 into the formula as the pmt. If pmt is omitted, you must include the fv argument.

Pv is the present value, or the lump-sum amount that a series of future payments is worth right now. If pv is omitted, it is assumed to be 0 (zero), and you must include the pmt argument.

Fv is the future value, or a cash balance you want to attain after the last payment is made. If fv is omitted, it is assumed to be 0 (the future value of a loan, for example, is 0). For example, if you want to save $50,000 to pay for a special project in 18 years, then $50,000 is the future value. You could then make a conservative guess at an interest rate and determine how much you must save each month. If fv is omitted, you must include the pmt argument.

Type is the number 0 or 1 and indicates when payments are due.

Set type equal to If payments are due
0 or omitted At the end of the period
1 At the beginning of the period

Guess is your guess for what the rate will be. (Only used in **RATE( )** function)
If you omit guess, it is assumed to be 10 percent.
If RATE does not converge, try different values for guess. RATE usually converges if guess is between 0 and 1.

Remarks

Make sure that you are consistent about the units you use for specifying rate and nper. If you make monthly payments on a four-year loan at 12 percent annual interest, use 12%/12 for rate and 4*12 for nper. If you make annual payments on the same loan, use 12% for rate and 4 for nper.

An annuity is a series of constant cash payments made over a continuous period. For example, a car loan or a mortgage is an annuity. For more information, see the description for each annuity function.

In annuity functions, cash you pay out, such as a deposit to savings, is represented by a negative number; cash you receive, such as a dividend check, is represented by a positive number. For example, a $1,000 deposit to the bank would be represented by the argument -1000 if you are the depositor and by the argument 1000 if you are the bank.

PMT(rate,nper,pv,fv,type)
Calculates the payment for a loan based on constant payments and a constant interest rate.

The following formula returns the monthly payment on a $10,000 loan at an annual rate of 8 percent that you must pay off in 10 months:

($1,037.03) <-- PMT(8%/12, 10, 10000)

For the same loan, if payments are due at the beginning of the period, the payment is:

($1,030.16) <-- PMT(8%/12, 10, 10000, 0, 1)

The following formula returns the amount someone must pay to you each month if you loan that person $5,000 at 12 percent and want to be paid back in five months:

$1,030.20 <-- PMT(12%/12, 5, -5000)

You can use PMT to determine payments to annuities other than loans. For example, if you want to save $50,000 in 18 years by saving a constant amount each month, you can use PMT to determine how much you must save. If you assume you'll be able to earn 6 percent interest on your savings, you can use PMT to determine how much to save each month.

($129.08) <-- PMT(6%/12, 18*12, 0, 50000)

If you pay $129.08 into a 6 percent savings account every month for 18 years, you will have $50,000.

FLAT PAYMENT SCHEDULES (PMT function for annuity calc)

=PMT(B4,B5,-B3)
[Payment Function]

Loan principal	10,000						
Interest rate	7%						
Loan term	6	<-- Number of years over which loan is repaid					
Annual payment	**2,097.96**	<-- To be made at end of each year					
LOAN TABLE ==>		**Principal**	**Payment**	**Split payment into:**			
	Year	**at begin.**	**at end of**		**Return of**		
		of year	**year**	**Interest**	**principal**		
	1	10,000.00	2,097.96	700.00	1,397.96		
	2	8,602.04	2,097.96	602.14	1,495.82		
	3	7,106.23	2,097.96	497.44	1,600.52		
	4	5,505.70	2,097.96	385.40	1,712.56		
	5	3,793.15	2,097.96	265.52	1,832.44		
	6	1,960.71	2,097.96	137.25	1,960.71		

=$B$4*C11

=D11-E11

=C11-F11

FV(Rate,Nper,Pmt,PV,Type)
Returns the future value of an investment based on periodic, constant payments and a constant interest rate.

1. FUTURE VALUE OF A SINGLE CF

FV of $1,000 in 5 years at 10%: **$1,610.51** <-- =FV(10%,5,0,-1000)

2. FUTURE VALUE WITH ANNUAL DEPOSITS (Annuity Due)

Interest 10%

Year	Account balance beg. year	Deposit at beginning of year	Interest earned during year	Total in account end of year	
0	0.00	1,000	100.00	1,100.00	<-- =E21+D21+C21
1	1,100.00	1,000	210.00	2,310.00	
2	2,310.00	1,000	331.00	3,641.00	=$C$16*(D21+C21)
3	3,641.00	1,000	464.10	5,105.10	
4	5,105.10	1,000	610.51	6,715.61	
5	6,715.61	1,000	771.56	8,487.17	
6	8,487.17	1,000	948.72	10,435.89	
7	10,435.89	1,000	1,143.59	12,579.48	
8	12,579.48	1,000	1,357.95	14,937.42	
9	14,937.42	1,000	1,593.74	17,531.17	
10	17,531.17				

=F21

Future value **$17,531.17** <-- =FV(C16,B31,-1000,,1)

PV(rate,nper,pmt,fv,type)
Returns the present value of an investment. The present value is the total amount that a series of future payments is worth now. For example, when you borrow money, the loan amount is the present value to the lender.

Suppose you're thinking of buying an insurance annuity that pays $500 at the end of every month for the next 20 years. The cost of the annuity is $60,000, and the money paid out will earn 8 percent. You want to determine whether this would be a good investment. Using the PV function, you find that the present value of the annuity is:

($59,777.15) <-- PV(0.08/12, 12*20, 500, , 0) **[Present Value of an Annuity]**

The result is negative because it represents money that you would pay, an outgoing cash flow. The present value of the annuity ($59,777.15) is less than what you are asked to pay ($60,000). Therefore, you determine this would not be a good investment.

Suppose that in addition to the $500 payment at the end of every month for the next 20 years, the insurance also pays $20,000 at the end of the 20th year. The cost of the annuity is $60,000, and the money paid out will earn 8 percent. You want to determine whether this would be a good investment. Using the PV function, you find that the present value of the annuity is:

($63,836.57) <-- =PV(0.08/12,12*20,500,20000,0)

Now this is a good investment because the PV is greater than the cost.

Suppose you will receive a lump-sum amount $200,000 20 years from now. The interest is 8% with monthly interest compounding. The present value of the amount can also be determined by PV( ) function.

($40,594.28) <-- PV(0.08/12, 12*20, ,200000) [Present Value of a future lump-sum]

NPER(rate, pmt, pv, fv, type)
Returns the number of periods for an investment based on periodic, constant payments and a constant interest rate.

NPER(1%, -100, -1000, 10000) equals 59.7 periods.

59.7 <-- NPER(12%/12, -100, -1000, 10000, 1)

NPER(1%, -100, 1000) equals 10.6 periods.

10.6 <-- NPER(1%, -100, 1000)

RATE(nper,pmt,pv,fv,type,guess)
Returns the interest rate per period of an annuity. RATE is calculated by iteration and can have zero or more solutions. If the successive results of RATE do not converge to within 0.0000001 after 20 iterations, RATE returns the #NUM! error value.

To calculate the rate of a four-year $8,000 loan with monthly payments of $200:

0.77%	<-- RATE(48, -200, 8000)

This is the monthly rate, because the period is monthly. The annual rate is 0.77%*12, which equals 9.24 percent.

9.24%	<-- =12*RATE(48, -200, 8000)

	A	B	C	D	E	F	G	H	I
1		**MULTIPLE COMPOUNDING PERIODS**							
2									
3		Initial deposit	1,000						
4		Interest rate	5%		**=C3*(1+C6)^C5**				
5		Number of compounding periods per year	2						
6		Interest per compounding period	2.500%		**=C3*EXP(C4)**				
7		Accretion in one year	**1050.625**						
8		**Continuous** compounding with **Exp**	**1051.271**						
9									
10									
11									
12		PV Calc with Continuous Compounding	Interest	8%					
13						Continuously			
14				Year	Cash flow	discounted PV			
15				1	100	92.312	**<-- =EXP(-$D$12*D15)*E15**		
16				2	200	170.429			
17				3	300	235.988			
18				4	350	254.152			
19				5	400	268.128			
20									
21				**Present value**		**1021.009**	**<-- =SUM(F15:F19)**		